// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/utils/ERC721Holder.sol)

pragma solidity ^0.8.0;

import "../IERC721Receiver.sol";

/**
 * @dev Implementation of the {IERC721Receiver} interface.
 *
 * Accepts all token transfers.
 * Make sure the contract is able to use its token with {IERC721-safeTransferFrom}, {IERC721-approve} or {IERC721-setApprovalForAll}.
 */
contract ERC721Holder is IERC721Receiver {
    /**
     * @dev See {IERC721Receiver-onERC721Received}.
     *
     * Always returns `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address,
        address,
        uint256,
        bytes memory
    ) public virtual override returns (bytes4) {
        return this.onERC721Received.selector;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: src/contracts/interfaces/ConduitInterface.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

enum ConduitItemType {
    NATIVE, // unused
    ERC20,
    ERC721,
    ERC1155
}

/**
 * @dev A ConduitTransfer is a struct that contains the information needed for a
 *      conduit to transfer an item from one address to another.
 */
struct ConduitTransfer {
  ConduitItemType itemType;
  address token;
  address from;
  address to;
  uint256 identifier;
  uint256 amount;
}

/**
 * @dev A ConduitBatch1155Transfer is a struct that contains the information
 *      needed for a conduit to transfer a batch of ERC-1155 tokens from one
 *      address to another.
 */
struct ConduitBatch1155Transfer {
  address token;
  address from;
  address to;
  uint256[] ids;
  uint256[] amounts;
}

/**
 * @title ConduitInterface
 * @author 0age
 * @notice ConduitInterface contains all external function interfaces, events,
 *         and errors for conduit contracts.
 */
interface ConduitInterface {
  /**
    * @dev Revert with an error when attempting to execute transfers using a
    *      caller that does not have an open channel.
    */
  error ChannelClosed(address channel);

  /**
    * @dev Revert with an error when attempting to update a channel to the
    *      current status of that channel.
    */
  error ChannelStatusAlreadySet(address channel, bool isOpen);

  /**
    * @dev Revert with an error when attempting to execute a transfer for an
    *      item that does not have an ERC20/721/1155 item type.
    */
  error InvalidItemType();

  /**
    * @dev Revert with an error when attempting to update the status of a
    *      channel from a caller that is not the conduit controller.
    */
  error InvalidController();

  /**
    * @dev Emit an event whenever a channel is opened or closed.
    *
    * @param channel The channel that has been updated.
    * @param open    A boolean indicating whether the conduit is open or not.
    */
  event ChannelUpdated(address indexed channel, bool open);

  /**
    * @notice Execute a sequence of ERC20/721/1155 transfers. Only a caller
    *         with an open channel can call this function.
    *
    * @param transfers The ERC20/721/1155 transfers to perform.
    *
    * @return magicValue A magic value indicating that the transfers were
    *                    performed successfully.
    */
  function execute(
      ConduitTransfer[] calldata transfers
  ) external returns (bytes4 magicValue);

  /**
    * @notice Execute a sequence of batch 1155 transfers. Only a caller with an
    *         open channel can call this function.
    *
    * @param batch1155Transfers The 1155 batch transfers to perform.
    *
    * @return magicValue A magic value indicating that the transfers were
    *                    performed successfully.
    */
  function executeBatch1155(
      ConduitBatch1155Transfer[] calldata batch1155Transfers
  ) external returns (bytes4 magicValue);

  /**
    * @notice Execute a sequence of transfers, both single and batch 1155. Only
    *         a caller with an open channel can call this function.
    *
    * @param standardTransfers  The ERC20/721/1155 transfers to perform.
    * @param batch1155Transfers The 1155 batch transfers to perform.
    *
    * @return magicValue A magic value indicating that the transfers were
    *                    performed successfully.
    */
  function executeWithBatch1155(
      ConduitTransfer[] calldata standardTransfers,
      ConduitBatch1155Transfer[] calldata batch1155Transfers
  ) external returns (bytes4 magicValue);

  /**
    * @notice Open or close a given channel. Only callable by the controller.
    *
    * @param channel The channel to open or close.
    * @param isOpen  The status of the channel (either open or closed).
    */
  function updateChannel(address channel, bool isOpen) external;
}

// File: src/contracts/protocols/SkilletProtocolBase.sol



//SPDX-License-Identifier: Skillet-Group
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol";

import "../interfaces/ConduitInterface.sol";

interface ISkilletRegistry {
  function conduitAddress() external view returns (address);
}

contract SkilletProtocolBase is ERC721Holder {
  using SafeERC20 for IERC20;
  uint256 MAX_UINT_256 = 2**256 - 1;

  address public skilletRegistryAddress;

  constructor(address _skilletRegistryAddress) {
    skilletRegistryAddress = _skilletRegistryAddress;
  }

  /**
   * @dev Get Conduit Address
   * @return conduitAddress address of the approved conduit
   */
   function getConduitAddress() public view returns (address conduitAddress) {
    conduitAddress = ISkilletRegistry(skilletRegistryAddress).conduitAddress();
   }

  /**
   * @dev Transfer asset from sender to reciever
   * @param sender address of the sender
   * @param receiver address of the receiver
   * @param collectionAddress address of the ERC721 collection
   * @param tokenId token identifier in collection
   */
  function transferERC721(
    address sender,
    address receiver,
    address collectionAddress,
    uint256 tokenId
  ) public 
  {
    IERC721 collection = IERC721(collectionAddress);
    collection.safeTransferFrom(sender, receiver, tokenId);
  }

  /**
   * @dev Transfers amout of currency from sender to recevier
   * @param receiver address of the receiver
   * @param sender address of the sender
   * @param currencyAddress address of the currency being transferred
   * @param paymentAmount amount of currency to be sent to receiver
   */
  function transferERC20 (
    address sender,
    address receiver,
    address currencyAddress,
    uint256 paymentAmount
  ) public
  {
    IERC20 currency = IERC20(currencyAddress);
    if (sender == address(this)) {
      currency.transfer(receiver, paymentAmount);
      return;
    }

    currency.transferFrom(sender, receiver, paymentAmount);
  }

  function approveOperatorForERC721(
    address operator,
    address collectionAddress
  ) public
  {
    IERC721 collection = IERC721(collectionAddress);
    collection.setApprovalForAll(operator, true);
  }

  function checkAndSetOperatorApprovalForERC721(
    address operator,
    address collectionAddress
  ) public
  {
    IERC721 collection = IERC721(collectionAddress);
    bool approved = collection.isApprovedForAll(address(this), operator);
    if (!approved) collection.setApprovalForAll(operator, true);
  }

  function approveOperatorForERC20(
    address operator,
    address currencyAddress
  ) public
  {
    IERC20 currency = IERC20(currencyAddress);
    currency.approve(operator, MAX_UINT_256);
  }

  function checkAndSetOperatorApprovalForERC20(
    address operator,
    address currencyAddress
  ) public
  {
    IERC20 currency = IERC20(currencyAddress);
    uint256 allowance = currency.allowance(address(this), operator);
    if (!(allowance == MAX_UINT_256)) currency.approve(operator, MAX_UINT_256);
  }

  function conduitTransferERC721(
    address collectionAddress,
    address from,
    address to,
    uint256 tokenId
  ) public
  {
    ConduitTransfer[] memory transfers = new ConduitTransfer[](1);
    transfers[0] = ConduitTransfer({
      itemType: ConduitItemType.ERC721,
      token: collectionAddress,
      from: from,
      to: to,
      identifier: tokenId,
      amount: 1
    });

    ConduitInterface(getConduitAddress()).execute(transfers);
  }

  function conduitTransferERC20(
    address currencyAddress,
    address from,
    address to,
    uint256 amount
  ) public
  {
    ConduitTransfer[] memory paymentTransfers = new ConduitTransfer[](1);
    paymentTransfers[0] = ConduitTransfer({
      itemType: ConduitItemType.ERC20,
      token: currencyAddress,
      from: from,
      to: to,
      identifier: 0,
      amount: amount
    });

    ConduitInterface(getConduitAddress()).execute(paymentTransfers);
  }
}

// File: src/contracts/protocols/x2y2/interfaces/IServiceFee.sol



// SPDX-License-Identifier: Skillet Group
pragma solidity ^0.8.0;

/**
 * Skillet <> X2Y2
 * Service Fee Interface
 * https://etherscan.io/address/0xb858E4a6f81173892AD263584aa5b78F2407EE72#code
 */
interface IServiceFee {
  function getServiceFee(address _target, address _sender, address _nftAsset) external returns (uint16);
}

// File: src/contracts/protocols/x2y2/interfaces/IXY3.sol



// SPDX-License-Identifier: Skillet Group
pragma solidity ^0.8.0;

/**
 * Skillet <> X2Y2
 * XY3 Interface
 * https://etherscan.io/address/0xFa4D5258804D7723eb6A934c11b1bd423bC31623#code
 */
interface IXY3 {

  struct Signature {
    uint256 nonce;
    uint256 expiry;
    address signer;
    bytes signature;
  }

  struct Offer {
    uint256 borrowAmount;
    uint256 repayAmount;
    address nftAsset;
    uint32 borrowDuration;
    address borrowAsset;
    uint256 timestamp;
    bytes extra;
  }

  struct CallData {
    address target;
    bytes4 selector;
    bytes data;
    uint256 referral;
  }

  struct LoanDetail {
    uint256 borrowAmount;
    uint256 repayAmount;
    uint256 nftTokenId;
    address borrowAsset;
    uint32 loanDuration;
    uint16 adminShare;
    uint64 loanStart;
    address nftAsset;
    bool isCollection;
  }

  function borrow(
    Offer calldata _offer,
    uint256 _nftId,
    bool _isCollectionOffer,
    Signature calldata _lenderSignature,
    Signature calldata _brokerSignature,
    CallData calldata _extraDeal
  ) external returns (uint32);

  function repay(uint32 loanId) external;

  function loanDetails(uint32 loanId) external view returns (LoanDetail memory);
}

// File: src/contracts/protocols/x2y2/interfaces/IXY3AddressProvider.sol



// SPDX-License-Identifier: Skillet Group
pragma solidity ^0.8.0;

/**
 * Skillet <> X2Y2
 * X2Y2 Address Provider Interface
 * https://etherscan.io/address/0x21A619115F36dE1A71B549e9081022fe84136f65#code
 */
interface IXY3AddressProvider {
  function getBorrowerNote() external view returns (address);
  function getTransferDelegate() external view returns (address);
  function getXY3() external view returns (address);
  function getServiceFee() external view returns (address);
}

// File: src/contracts/protocols/x2y2/SkilletxX2Y2.sol



//SPDX-License-Identifier: Skillet-Group
pragma solidity ^0.8.0;

import "../SkilletProtocolBase.sol";
import "./X2Y2AddressProvider.sol";
import "./X2Y2ServiceFee.sol";
import "./X2Y2Originator.sol";
import "./X2Y2Utils.sol";

import "./interfaces/IXY3.sol";

import "./structs/X2Y2Structs.sol";

/**
 * Skillet <> X2Y2
 * https://etherscan.io/address/0xfa4d5258804d7723eb6a934c11b1bd423bc31623#code
 */
contract SkilletxX2Y2 is 
  SkilletProtocolBase,
  X2Y2AddressProvider,
  X2Y2ServiceFee,
  X2Y2Originator,
  X2Y2Utils
{

  constructor(address _skilletRegistryAddress) SkilletProtocolBase(_skilletRegistryAddress) {}

  event X2Y2LoanStarted(
    address indexed borrower, 
    uint256 indexed loanId, 
    uint256 indexed obligationId,
    address collectionAddress,
    uint256 tokenId,
    uint256 netAmountOwed
  );

  event X2Y2LoanClosed(
    address indexed borrower, 
    uint256 indexed loanId, 
    uint256 indexed obligationId,
    address collectionAddress,
    uint256 tokenId,
    uint256 netAmountDue
  );

  event X2Y2LoanRolledOver(
    address indexed borrower, 
    uint256 indexed oldLoanId, 
    uint256 indexed newLoanId
  );

  /**
   * Start a new X2Y2 Loan
   * @param loan a structured X2Y2 loan object
   * @param borrowerAddress address of the borrower account
   * @param collectionAddress address of the ERC721
   * @param tokenId identifier of the ERC721
   */
  function startX2Y2Loan(
    X2Y2Loan calldata loan,
    address borrowerAddress,
    address collectionAddress,
    uint256 tokenId
  ) public 
  {

    // transfer ERC721 to contract
    conduitTransferERC721(
      collectionAddress,
      borrowerAddress,
      address(this),
      tokenId
    );

    // check and set approval for DirectLoanFixedCollectionOffer
    checkAndSetOperatorApprovalForERC721(
      addressProvider.getTransferDelegate(), 
      collectionAddress
    );
    
    // start loan offer
    uint32 loanId = borrow(loan, tokenId);

    // get borrowerNoteId
    uint256 borrowerNoteId = getBorrowerNoteId(loanId);

    // send principal to borrower
    transferERC20(
      address(this),
      borrowerAddress,
      loan.offer.borrowAsset,
      loan.offer.borrowAmount
    );

    // transfer obligation receipt to borrower
    transferERC721(
      address(this),
      borrowerAddress,
      addressProvider.getBorrowerNote(),
      borrowerNoteId
    );

    emit X2Y2LoanStarted(
      borrowerAddress, 
      uint256(loanId), 
      borrowerNoteId,
      collectionAddress,
      tokenId,
      loan.offer.borrowAmount
    );
  }

  /**
   * Repay an existing X2Y2 Loan
   * @param loanId identifier of the outstanding loan
   * @param borrowerAddress address of the borrower account
   * @param collectionAddress address of the ERC721
   * @param tokenId identifier of the ERC721
   */
  function repayX2Y2Loan(
    uint32 loanId,
    address borrowerAddress,
    address collectionAddress,
    uint256 tokenId
  ) public
  {

    // get borrowerNote id from loanId
    uint256 borrowerNoteId = getBorrowerNoteId(loanId);

    // transfer borrowerNote to contract
    conduitTransferERC721(
      addressProvider.getBorrowerNote(),
      borrowerAddress,
      address(this),
      borrowerNoteId
    );

    // get loan information
    IXY3.LoanDetail memory _loanDetails = getLoanDetails(loanId);

    // transfer currency to contract
    conduitTransferERC20(
      _loanDetails.borrowAsset,
      borrowerAddress,
      address(this),
      _loanDetails.repayAmount
    );

    // approve x2y2 delegate
    approveOperatorForERC20(
      addressProvider.getTransferDelegate(),
      _loanDetails.borrowAsset
    );

    // repay loan
    repay(loanId);

    // transfer asset back to borrower
    transferERC721(
      address(this),
      borrowerAddress,
      collectionAddress,
      tokenId
    );

    emit X2Y2LoanClosed(
      borrowerAddress, 
      uint256(loanId), 
      borrowerNoteId,
      collectionAddress,
      tokenId,
      _loanDetails.repayAmount
    );
  }

  /**
   * Rollover an X2Y2 loan into a new loan
   * @param loan a structured X2Y2 loan object
   * @param borrowerAddress address of the borrower account
   */
  function rolloverX2Y2Loan(
    X2Y2Loan calldata loan,
    address borrowerAddress
  ) public
  {

    // decode the extraDeal data into the oldLoanId
    uint32 oldLoanId = abi.decode(loan.extraDeal.data, (uint32));

    // get borrowerNoteId from the oldLoanId
    uint256 borrowerNoteId = getBorrowerNoteId(oldLoanId);
    
    // transfer the borrowerNote to the contract
    conduitTransferERC721(
      addressProvider.getBorrowerNote(),
      borrowerAddress,
      address(this),
      borrowerNoteId
    );

    // get loan information
    IXY3.LoanDetail memory _loanDetails = getLoanDetails(oldLoanId);

    // approve x2y2 delegate
    approveOperatorForERC20(
      addressProvider.getTransferDelegate(),
      _loanDetails.borrowAsset
    );

    // calculate total rollover fee
    uint256 totalServiceFee = calculateServiceFee(
      loan.offer.borrowAmount,
      _loanDetails.nftAsset
    );
    
    // transfer amount due for repayment from user to contract
    if (loan.offer.borrowAmount < _loanDetails.repayAmount + totalServiceFee) {
      uint256 paymentDue = _loanDetails.repayAmount + totalServiceFee - loan.offer.borrowAmount;
      conduitTransferERC20(
        _loanDetails.borrowAsset,
        borrowerAddress,
        address(this),
        paymentDue
      );
    }
    
    // start new loan offer
    uint32 newLoanId = borrow(loan, _loanDetails.nftTokenId);

    // get borrowerNoteId
    uint256 newBorrowerNoteId = getBorrowerNoteId(newLoanId);

    // transfer new borrowerNote to borrower
    transferERC721(
      address(this),
      borrowerAddress,
      addressProvider.getBorrowerNote(),
      newBorrowerNoteId
    );

    // transfer amount owed from net principal from contract to the user
    if (loan.offer.borrowAmount > _loanDetails.repayAmount + totalServiceFee) {
      uint256 paymentOwed = loan.offer.borrowAmount - (_loanDetails.repayAmount + totalServiceFee);
      transferERC20(
        address(this),
        borrowerAddress,
        _loanDetails.borrowAsset,
        paymentOwed
      );
    }

    emit X2Y2LoanRolledOver(
      borrowerAddress,
      uint256(oldLoanId),
      uint256(newLoanId)
    );
  }
}

// File: src/contracts/protocols/x2y2/structs/X2Y2Structs.sol



// SPDX-License-Identifier: Skillet Group
pragma solidity ^0.8.0;

import "../interfaces/IXY3.sol";

struct X2Y2Loan {
  IXY3.Offer offer;
  IXY3.Signature lenderSignature;
  IXY3.Signature brokerSignature;
  IXY3.CallData extraDeal;
  bool isCollectionOffer;
}

// File: src/contracts/protocols/x2y2/X2Y2AddressProvider.sol



//SPDX-License-Identifier: Skillet-Group
pragma solidity ^0.8.0;

import "./interfaces/IXY3AddressProvider.sol";

/**
 * Skillet <> X2Y2
 * Address Provider
 * https://etherscan.io/address/0x21A619115F36dE1A71B549e9081022fe84136f65#code
 */
contract X2Y2AddressProvider {
  address public x2y2AddressProviderAddress = 0x21A619115F36dE1A71B549e9081022fe84136f65;
  IXY3AddressProvider public addressProvider = IXY3AddressProvider(x2y2AddressProviderAddress);

  function getBorrowerNoteAddress() public view returns (address borrowerNoteAddress) {
    borrowerNoteAddress = addressProvider.getBorrowerNote();
  }

  function getTransferDelegateAddress() public view returns (address transferDelegateAddress) {
    transferDelegateAddress = addressProvider.getTransferDelegate();
  }
}

// File: src/contracts/protocols/x2y2/X2Y2Originator.sol



//SPDX-License-Identifier: Skillet-Group
pragma solidity ^0.8.0;

import "./interfaces/IXY3.sol";
import "./X2Y2AddressProvider.sol";

import "./structs/X2Y2Structs.sol";

/**
 * Skillet <> X2Y2
 */
contract X2Y2Originator is X2Y2AddressProvider {

  function getLoanAddress() public view returns (address loanAddress) {
    loanAddress = addressProvider.getXY3();
  }

  function getLoanDetails(
    uint32 loanId
  ) public 
    view 
    returns (IXY3.LoanDetail memory loanDetails) 
  {
    IXY3 originator = IXY3(addressProvider.getXY3());
    loanDetails = originator.loanDetails(loanId);
  }

  function borrow(
    X2Y2Loan memory loan,
    uint256 tokenId
  ) public
    returns (uint32 loanId)
  {
    IXY3 originator = IXY3(addressProvider.getXY3());
    loanId = originator.borrow(
      loan.offer,
      tokenId,
      loan.isCollectionOffer,
      loan.lenderSignature,
      loan.brokerSignature,
      loan.extraDeal
    );
  }

  function repay(
    uint32 loanId
  ) public
  {
    IXY3 originator = IXY3(addressProvider.getXY3());
    originator.repay(loanId);
  }
}

// File: src/contracts/protocols/x2y2/X2Y2ServiceFee.sol



//SPDX-License-Identifier: Skillet-Group
pragma solidity ^0.8.0;

import "./interfaces/IServiceFee.sol";
import "./X2Y2AddressProvider.sol";

/**
 * Skillet <> X2Y2
 * Service Fee Implementation
 * https://etherscan.io/address/0xb858E4a6f81173892AD263584aa5b78F2407EE72
 */
contract X2Y2ServiceFee is X2Y2AddressProvider {
  function calculateServiceFee(
    uint256 amount,
    address collectionAddress
  ) internal
    returns (uint256)
  {
    IServiceFee serviceFeeController = IServiceFee(addressProvider.getServiceFee());
    uint16 serviceFee = serviceFeeController.getServiceFee(
      addressProvider.getXY3(),
      address(this),
      collectionAddress
    );

    return amount * serviceFee / 10000;
  }
}

// File: src/contracts/protocols/x2y2/X2Y2Utils.sol



//SPDX-License-Identifier: Skillet-Group
pragma solidity ^0.8.0;

import "./X2Y2AddressProvider.sol";

contract X2Y2Utils is X2Y2AddressProvider {

  function getBorrowerNoteId(
    uint32 loanId
  ) public 
    view 
    returns (uint256 borrowerNoteId)
  {
    borrowerNoteId = uint256(
      uint64(
        uint256(keccak256(abi.encodePacked(addressProvider.getXY3(), loanId)))
      ));
  }
}